
Abstract 
 
This is a report on my methods, the resources, and my tools for the support of users, software, 
and a BI server. All of the tools were built with SAS software itself. Code for the tools is either 
in the paper itself, or in references cited therein. 
 
Introduction 
 
This paper is structured as three chapters. The first chapter is mainly a list of the kinds of 
activities that a person who supports a SAS server (EBI, BI, or simple) might do to maximally 
support the environment. That chapter mentions some support tools. The remainder of the paper 
is devoted to those tools. The second chapter is an improved republication of my SAS Global 
Forum 2010 paper “More Ways to Use SAS to Manage, Monitor, and Control SAS or the SAS 
BI Server: Tools for the SAS User, Server Administrator, or Manager”, which is an expansion of 
my SAS Global Forum 2009 paper, which had a similar title, but a smaller scope. The third 
chapter describes a web-enabled environment that can be used to support SAS users. The code to 
build that environment can be requested via email as a zip file that will also include all of the 
files that support a model environment.

Chapter 1. How To Support Users of a SAS Server (or Stand-alone PC SAS) 
 
The fact of the matter is that often at SAS sites someone from IT installs SAS on a server, or on 
PCs, and will apply upgrades or fixes from time to time, but SAS users are otherwise on their 
own. Frankly, there probably are some SAS users who actually prefer it that way, and other SAS 
users who don’t realize that it could be any other way. For a SAS BI Server or a SAS EBI 
Server, and whatever might be the SAS vendor’s latest preferred way to describe the most 
current server architecture, there are some mandatory server administration duties that users are 
not permitted to perform for themselves. 
 
For six-and-a-half years, I supported a SAS BI server. Below is a list of what I did. 
 
 supported the SAS BI server 
 supported the SAS software 
 served as the essential subject matter expert for IT personnel on all problems and matters 

involving the SAS server, including performance, tuning, troubleshooting, gathering and 
interpretation of diagnostics, and decisions about configuration 

 coordinated the installation of upgrades, enhancements, and changes to SAS software 
 performed testing and quality assurance for all upgrades, enhancements, and changes to SAS 

software, including new releases, hot fixes, and service packs 
 performed SAS testing and SAS quality assurance for any upgrades, enhancements, and 

changes to non-SAS-software aspects of the hardware, software, and network that affected or 
involved the SAS server 

 scheduled and coordinated all planned outages on the SAS server 
 alerted users affected by unplanned outages on the SAS server, notifying them if there was to 

be an emergency reboot, and keeping them informed as to status and progress during the 
outages 

 reviewed, evaluated, and selected all SAS software fixes for relevance 
 assisted and served as user liaison/educator for upgrades to SAS server and SAS client 

software 
 developed monitor to track who was using the SAS server, when, and with what impact on 

CPU and memory resources 
 developed monitor with email alerts to notify users whenever their SAS process had 

consumed a threshold amount of CPU time, with follow-up alerts upon further consumption 
of each threshold increment, with email alert CC’s to myself as Server Administrator 

 developed tools to capture, parse, and analyze SAS-related records from the Windows Event 
Log 

 developed disk space usage reporting for permanent data, with emails to users of shared disks 
that reached or exceeded a threshold 

 developed monitor to detect old data not cleaned up from critical shared disk used for SAS 
work data sets, sending email alert to owning user, with CC to myself as Server 
Administrator—if the shared disk fills up, an entire group of users is immediately denied 
service 

 served as liaison to SAS vendor on all aspects of the SAS software, licensing, and cost 
matters 

 reviewed and advised on annual renewal of licensing for SAS software portfolio 
 served as liaison and coordinator for all SAS-provided on-site training  
 designed the BI server user groups and corresponding logical servers 
 assisted configuration of the SAS/IntrNet server environment  
 developed the proof of concept demonstration application for SAS/IntrNet  
 supported all SAS-server-hosted data, regardless of type 
 defined and administered Windows security to control access to SAS-server-hosted data 
 performed disk space planning 
 allocated disk space for applications and users 
 developed the prototype applications for routine scheduled downloads from Oracle to the 

SAS analytical database 
 ported outside vendor data (such as that of Acxiom) to the SAS analytical database—porting 

required rather complex manipulation of the raw input to support a maximally usable 
environment of SAS data sets, SAS format libraries for code-to-text translations, and SAS 
labels for data description; validated accuracy of load  

 implemented connectivity of the SAS server to Microsoft SQL Server databases 
 supported all SAS users 
 supported users of client tools SAS Enterprise Guide and SAS Add-in for Microsoft Office 
 supported users of SAS in Batch 
 developed and executed surveys of SAS user satisfaction and SAS user needs 
 organized and hosted internal SAS users meetings for user mutual education and networking 
 administered SAS BI server user access definition and controls 
 helped SAS users on matters of “how to”, performance, or malfunction 
 performed SAS training 
 developed tools so that users could monitor, control, and audit their processing 
 built and maintained custom macros for general and special use 
 prepared, disseminated, and internal-web-posted documentation on all facilities and matters 

of interest to SAS users 
 designed, created (using SAS and ODS), and maintained a web-based User Documentation 

and Tools environment for guided and explained access to local documentation and 
resources, to tips & news, and to SAS information resources around the world 

 provided web-based guidance to users so that they could get immediate telephone help from, 
or online-submit problems to, SAS Technical Support—as option when he was unavailable 
as first level of support 

 
From 1985 to 1996, I supported SAS users on a mainframe. SAS was simpler then. There was no 
BI or EBI architecture. But for mainframe SAS it was possible to monitor who was using exactly 
which SAS PROCs, and therefore which SAS components. Besides some of the same activities 
as above, I installed and maintained the mainframe SAS software myself, and did the following: 
 
 extracted SAS SMF records to create a database of information about which users were using 

which SAS PROCs (and, by inference, which SAS products)  
 used Base SAS to develop an interactive tool to interrogate that database

Abstract 
 
This paper adds two tools to last year’s kit. 
 
The CPUmon tool sends emails to the SAS server user, with CC to the administrator or any other designee, 
whenever the user’s SAS process exceeds a CPU time threshold and sends a further email to the user whenever a set 
CPU increment is exceeded. 
 
The LogTimer SAS user macro puts in the SAS log the elapsed time and CPU time consumed between invocation 
with parameter Start and End. Without it, SAS provides only step-level numbers and a datetime for start of a Display 
Manager or SAS Enterprise Guide® session. Code resubmission does not change that datetime. SAS 9.2 system 
option FULLSTIMER adds a step-end datetime in the log. The LogTimer macro does not require the option and 
works in all current SAS versions. 
 
The 2009 predecessor to this paper, whose content is included here, met other needs. 
 
The 2009 tools answered questions for the administrator or manager of the server. Who is using SAS now and how 
much CPU time and memory? Since when? What is the last time each user was on the server? How heavily does 
each use the server in terms of frequency or CPU time? 
 
The 2009 tools helped the SAS user with information and empowerment. What processes do I myself have running? 
I have a possible looping or hung process on the remote server, but my SAS Enterprise Guide session is hung or 
cancelled. How can I kill the process? 
 
All of the tools, 2009 and 2010, were developed for a Windows BI server using SAS software. They can also be 
used for SAS on a stand-alone PC. 
 
Introduction 
 
Intended Audience 
 
The intended audience for the user monitoring and CPU-time monitoring tools is SAS BI Server administrators and 
possibly their managers. UserMon and CPUmon can not be implemented or operated by users of the BI Server. If 
their data is made read-accessible to users, they, like an administrator, can run queries against the UserMon data, 
including data pertaining to other users, unless you filter it first. (Though not the intended purpose for which these 
monitoring tools were invented, they could be adapted to monitor any non-SAS process(es) as well or instead.) 
 
The LogTimer, ShowProcessID, DisplayAllMySASprocesses, and TerminateProcess macros are intended for any 
SAS server user or PC SAS user.  
 
The BI Server unrestricted user, presuming that she/he has administrator rights on the server, can deal with process 
display and termination for anyone’s process on a server with DOS commands, rather than have to open SAS and 
run some macros. 
 
Why I Did This 
 
I got into the SAS Enterprise Guide client / remote SAS server architecture with SAS Enterprise Guide 2.5 and SAS 
8.2 in 2003. When I first read about the SAS 9.1.3 BI Server, I was excited by the prospect of a “SAS Management 
Console”. 
 



Chapter 2. Tools to Manage, Monitor, and Control SAS or the SAS BI Server 

5 

I expected a tool that would enable me, as server administrator, to see who was using the server and their resource 
burden, and to find and, if needed, kill hung, looping, or abandoned processes. The real function of the Version 9.1.3 
SAS Management Console is administration of a metadata repository to define BI server resources and control 
access to them. The SAS Management Console does not enable control or monitoring of the BI server operations. 
 
So, I developed my own server monitoring and control tools. Of course, there are various vendor-provided monitors 
for Windows servers. You might have something that you like better than what I can show you. If you have personal 
control over the configuration of such a tool, over what logging it does, and over how the log data, if any, is 
managed and made available to you, perhaps you are satisfied. If not, then this paper can help you. 
 
The UserMon monitor captured enough for my needs, and the configuration, logging, and data management were 
customizable to my personal preference, and were changeable by me at will. 
 
Because it is not a vendor-provided, more limited tool, my CPUmon monitor has the ability to translate the user ID 
associated with any process that is over the limit into the user’s email address and name. This information (and even 
the phone number) can be stored in the BI Server metadata database, or anywhere else that you prefer. In actual 
production use, my CPUmon did a lookup in a support table that was an extract from the metadata. Though I wanted 
to be notified about over-the-limit processes, I also wanted the process-owning user to be notified directly. I could 
not be continuously paying attention, but the user should be, and, in any case, is the responsible party. Of course, 
after every update by me to the BI server’s user metadata, I needed to refresh the extract lookup table. 
 
Scope and Structure of the Paper 
 
The server operating system is Windows. I don’t know whether UNIX analogues to all of these tools can be 
developed, but I can report that another user developed a UNIX tool to serve the function of my TerminateProcess 
macro. My tools specifically rely on Windows commands. The only SAS product needed is Base SAS. Macro 
language is used. ODS can be used to format reports of monitor data. Macro language and ODS are in Base SAS. 
 
Though I present results of an example query against the UserMon data here, answering all of the questions listed in 
the Abstract is left as a programming exercise for the reader who is an interested user.  
 
UserMon captures user ID (misleadingly called UserName by the Windows). More useful reports include the actual 
name of the user, which can, e.g., be obtained from an export of the user information sector of the BI Server 
metadata database or from any other lookup table maintained to provide the same information in a non-BI-server 
environment. 
 
A Tool That Everyone Can Use: The LogTimer Macro 
 
Q: When did my SAS program run, and what were its elapsed time and total CPU time? 
 
A: Though the SAS log displays the real time (elapsed time) and the CPU time for each DATA step and each PROC 
step, it does not answer the questions of when your program ran, and, unless it was run in batch, what the elapsed 
time and CPU time were for the entire block of code submitted, which is usually multi-step. 
 
The LogTimer macro is an easy-to-use tool that answers those questions. Without it, the only timing content in the 
SAS log, from SAS Display Manager or SAS Enterprise Guide, is of this form: 
 
NOTE: DATA statement used (Total process time): 
real time 0.41 seconds 
CPU time 0.02 seconds 
 
If your program runs in batch, step-level information like that above will be provided, but also the SAS log will end 
with a NOTE of this form: 
 
NOTE: The SAS System used: 
real time 16.13 seconds 

CPU time 2.06 seconds 
 
In SAS 9.2, if you turn on the system option FULLSTIMER, your SAS log will display the date/time at the end of 
each step. The LogTimer macro works in all versions of SAS, and does not require FULLSTIMER to be turned on. 
Before the SAS log content for the execution of the main body of your code, the LogTimer macro inserts something 
like this: 
 
***************************************** 
Started at 4:42:02 on 27OCT2009 
***************************************** 
 
and after the SAS log content for the execution of the main body of your code, the LogTimer macro inserts 
something like this: 
 
***************************************** 
Ended at 4:42:04 on 27OCT2009 
Elapsed Time (hours:minutes:seconds) = 0:00:02 
CPU Time (hours:minutes:seconds) = 0:00:01 
***************************************** 
 
Here is how to use the LogTimer macro: 
 
%LogTimer(Start); 
<your code here> 
%LogTimer(End); 
 
Invocation of the macro with “End” before “Start” or invocation of the macro with any other value will result in an 
appropriate error message. 
 
The “End” invocation of the LogTimer macro can be used, if desired, at multiple points within a code submission, in 
which case it, of course, displays the cumulative Elapsed Time and CPU Time as processing progresses. Discrete 
Elapsed Time and CPU Time for each constituent DATA step or PROC step are provided by SAS itself in the usual 
manner. 
 
NOTE: See Appendix 1 for the code for this macro. As stated in the macro’s internal comments, some parts of the 
code which captures CPU time are dependent on the specific version of Windows on which the macro is running. 
 
Tools for What the SAS Enterprise Guide Facilities Do Not Help With 
 
Purpose 
 
The TerminateProcess macro is intended for when your process is in one of these situations: 
 
Case 1. Your SAS Enterprise Guide session is no longer available, but the process is still running. This can happen if 
your PC/laptop has been rebooted or shut down without closing the SAS Enterprise Guide session, or if you used 
Windows Task Manager to kill SAS Enterprise Guide in desperation because it was frozen. 
 
Case 2. Your SAS Enterprise Guide session is open, but the red square Stop button and the Task Status Window do 
not terminate your running process, and you can not wait for normal termination (or do not think normal termination 
will ever occur). 
 
SAS provides no way to list all of your processes on the server. You can use the DisplayAllMySASprocesses macro 
at any time. E.g., you might want to know how much CPU time your puzzlingly long-running process has used up 
so far. Is it using an excessive amount? I.e., is it possibly in a loop? Is it getting NO CPU time? I.e., is someone 
else's process consuming all of the resources, or is your process waiting for something to happen? You can not 
display other people's processes. But a query to the UserMon data base can do that for you. 

How To Use Them 
 
These tools are intended to be used in the SAS Enterprise Guide Code Window.  However, the ShowProcessID 
macro can also be used at the beginning of a batch job whose log you are monitoring during its execution. 
 
1. Using the ShowProcessID Macro 
 
In the Code Window, submit this statement: 
%ShowProcessID; 
 
Look in the Log Window for feedback. You will get a message of this form: 
 
Process ID for this SAS Enterprise Guide session or SAS batch job is 6212 
 
2. Using the DisplayAllMySASprocesses Macro 
 
In the Code Window, submit this statement: 
 
%DisplayAllMySASprocesses; 
 
In the Log Window, look for a listing of your processes (where PID is Process ID): 
 
====== ============ ============ 
PID    Memory Usage CPU Time 
====== ============ ============ 
884        12,268 K      0:01:24 
3368        6,452 K      0:01:39 
====== ============ ============ 
 
3. Using the TerminateProcess Macro 
 
(This must be used in an SAS Enterprise Guide session other than the one of the process that you want to 
terminate.) In the Code Window, submit this statement: 
 
%TerminateProcess(ProcessID=NNNNNN); 
 
where NNNNNN is the Process ID (a.k.a., “PID”) for the process ID that you want to terminate. 
NNNNNN is typically a three- to six-digit number. 
 
Look in the Log Window for feedback. If your request succeeded, you will get this message: 
 
SUCCESS: Sent termination signal to the process with PID NNNNNN. 
 
Note that it states that a termination signal was Sent. To verify actual termination, resubmit 
%DisplayAllMySASprocesses;  
 
If you inadvertently try to kill a process that is not yours, the action will fail, and you will get this message 
(where ABCDEFGH will be your user ID): 
 
Process ID NNNNNN is not for User ID ABCDEFGH and will not be killed. 
 
If you inadvertently try to kill a non-existent process, the action will fail, and you will get this message: 
 
Process ID NNNNNN was not found. 
 
NOTE: See Appendix 1 for the code for these macros. 

Using SAS to Monitor SAS Users and Processes 
 
The heart of these tools is a DOS tasklist command 
 

tasklist /v /fi "imagename eq sas.exe" 
 
issued from a SAS program running on the server with a user ID that has the credentials of a Windows System 
Administrator. The BI Server unrestricted user might have such credentials. /v stands for “verbose”. 
 
This command could be used to find users and processes for any executable, not just SAS, by eliminating the filter: 
 

/fi "imagename eq sas.exe" 
 
The mechanism for data capture is a program that wakes up every six minutes (you can adjust this to any interval 
you prefer) to issue the command above with its output directed to a named pipe, followed by a DATA step that 
converts the output to a datetimestamped SAS data set that is written to permanent disk. The logic that parses the 
data returned in response to the tasklist command is Windows-version-dependent. 
 
Each monitor runs as a Windows Scheduled Task, using a bat file of this form: 
 
"C:\Program Files\SAS\SAS 9.1\sas.exe"  -sysin "PathToPgm\UserMon.sas"  -log 
"PathToLog\UserMonSASlog.txt"  -print "PathToLst\UserMonSASlst.txt"  -work "PathToWork" 
 
where PathToPgm, PathToLog, PathToLst, and PathToWork could all be assigned instead as the same Windows 
folder. Also, each monitor could run as a Windows service that starts automatically at reboot. 
 
The monitors are not a big resource burden. Their CPU time and disk space needs are not significant. 
 
You can run either monitor without running the other.  
 
UserMon 
 
Here are excerpts from two of many reports that can be created from the UserMon data: 
 

   
For each monitor interval, UserMon creates a datetime-suffixed SAS data set of records, one record for each Process 
ID, along with the User ID, Cumulative CPU seconds, Current Memory Usage, and Monitor DateTime for that 
Process ID. To work with the monitor data, you need to concatenate all of the monitor-datetime-specific SAS data 

sets into one cumulative data set, optionally subject to start and end day filtering. If you do not cut off data 
concatenation at a day earlier than the concatenation run day, it is important to avoid the latest data set since the 
SAS user monitor might be awake and actively writing to it. 
 
NOTE: See Appendix 1 for code used for the monitor, to concatenate monitor logs, and to create these reports. 
 
CPUmon 
 
At each monitor interval, CPUmon looks for any SAS processes that have consumed more than two CPU hours (by 
default), or whatever has been set as the cumulative CPU time threshold. The UserID, Process ID (a.k.a. “PID”), and 
CPU time are written to a reference table for any alert. Whenever such a process is detected, the table is checked for 
a pre-existing record. If no record is found, the alert is sent, and an alert record is created in the table. If there is a 
pre-existing record, and the current cumulative CPU time is over the previously alerted CPU time by a defined 
increment (one hour by default), another alert is sent, and the new CPU time is recorded in that row of the table. The 
increment trigger for additional alerts, like the threshold trigger for the initial alert, is a macro invocation parameter. 
Any PIDs that are no longer active at the monitor instant are deleted from the table. Alerts are sent directly to the 
user, with CC to an administrator email address. If no email address is available for the user, both TO and CC use 
the administrator email address. The administrator email address is a macro invocation parameter, but, of course, 
with a code change to the macro it could be accessed from a lookup file instead. 
 
As a point of programming information, the reference table actually is not updated in place. It is rebuilt during every 
monitor interval, retaining only fresh alerts, old alerts still running, or old alerts still running for which a renewed 
alert has been sent. 
 
The macro supplied in this paper does include optional code that can be turned on to permit inspection of the 
response from the tasklist command. You need to customize it to write the diagnostic information to a location 
where the monitor will, in fact, have WRITE authorization. 
 
The layout of the response can vary by version of Windows, and the code needs to be revised to suit the 
environment. Also, structure and length of user ID will vary from site to site. And each site will have its own way of 
wanting to support the translation of user ID into email address. 
 
Finally, the macro includes the capability to send a quite verbose email message. It is designed to be sent to users of 
a server. (It could be used on a standalone PC.) By modifying the macro, the message can be customized to fit the 
actual characteristics of the installed site and to suit the preferences of the site SAS Administrator. The approach 
taken in the code for the email message here is that a maximally concise message, especially the first time received, 
could be confusing to a user who is not aware of the implications of using a lot of CPU time. Also, in some cases, 
using a lot of CPU time is expected and harmless.  
 
NOTE: See Appendix 1 for the CPUmon macro and an example of its invocation. If adapting it to your site, you 
need to verify the parsing of the TaskList response, customize the email message, customize the email address 
lookup, possibly remove the diagnostic code, and probably simplify or remove the ReportTimeUnits handling 
(which was designed to support a variety of testing scenarios). 
 
Below is an example of the email alert sent to a fictitious SAS user. The recipient email address, User ID, Process 
ID, name of the user, server name, date, and time are bogus. 
 
From: LeRoy Bessler [mailto:Le_Roy_Bessler@wi.rr.com]  
Sent: Sunday, February 28, 2010 11:59 PM 
To: NoSuchUser@NoneSuch.com 
Cc: Le_Roy_Bessler@wi.rr.com 
Subject: User ID CPUtestUserID Process 314159 Has Used 2.01 Hours of CPU Time. Is it OK? 
 
Dear FirstName LastName 
  
Please read this long automated message carefully and in entirety. 

As Of 28FEB2010:23:59:03.45 
User ID CPUtestUserID Process 314159 Has Used 2.01 Hours of CPU Time. 
Is it OK? If not, please terminate it. 
  
For information on how to terminate a SAS process, go to 
the URL for that part of the SAS Server Site User Documentation. 
If you can not terminate it, call the SAS Server Administrator. 
If the SAS Server Administrator is unavailable, call the Help Desk. 
They need the Process ID 314159 and Server Name IamYourSASservant. 
  
You will get a message every time current process 314159 uses another 1.00 Hours of CPU time. 
The process ID 314159 was assigned when your SAS Enterprise Guide or Display Manager session or batch 
job started. 
It varies from session to session and from job to job. 
  
CPU time is NOT elapsed time. It is active processing time. 
The CPU waits idle for file reads, file writes, responses from Oracle, etc. 
CPU time is usually less than elapsed time. 
 
Using a lot of CPU time is not necessarily bad. 
Heavy-duty analytics processes can use a lot of CPU time. 
However, if your process is in a loop, it is wasting resources. 
If your process is from an abandoned or frozen Enterprise Guide session, 
you will be unaware of how much CPU time it is using 
unless you investigate with the process management tools, 
or it has used at least 2 Hours of CPU time and you get this message. 
If you have any questions, call or send email to the SAS Server Administrator. 
 
Conclusion 
 
SAS can be used to put more run-related information in its SAS log and to monitor and control the SAS BI server. 
To support the server, and to further empower server users, I have developed other tools, not all of which could be 
presented here. One of the tools not presented is EGbatch. 
 
EGbatch allows users to submit SAS code from Enterprise Guide, but gives them the ability to monitor the SAS log 
on disk while their code is running. Via email, it sends the user a job start, a job end, and a job completion status 
message. The start and end messages include Process ID for the job. All three messages identify the SAS code file 
run by the job. In cases where the log size is not a concern for email, there is the option to have it attached to the job 
completion status email whenever there is an abnormal termination. Any report that the code might create can be 
emailed to a list of recipients. The ODS packaging options of normal SAS Enterprise Guide submissions are 
supported by EGbatch with no extra coding required in the submitted code itself. If interested, send me an email.  
 
I always enjoy finding ways to make SAS software do something that it does not do on its own accord. Being a 
long-time SAS enthusiast, I like to say, “If you can’t do it with SAS software, maybe you don’t really need to do it.” 

Contact Information, Etc. 
 
Your comments, questions, and suggestions are welcome. All code presented here is available in a zip file upon 
email request to the author. If you have any better ideas, or alternative ideas, as long as they do not require extra 
software and do not require non-SAS coding (other than DOS commands), I would be interested to see them. 
 
LeRoy Bessler PhD                                                                         Zum sehen geboren, zum schauen bestellt. 
Le_Roy_Bessler@wi.rr.com                                                           (Born to see, meant to look.) 

    —Goethe 
Strong Smart Systems™      
Visual Communication Power™      
 
A SAS user since 1978, Dr. LeRoy Bessler has shared his knowledge and experience with other users at conferences 
throughout the USA and in Montreal, London, Heidelberg, and Dublin. Though a SAS generalist with long 
experience in Base SAS, SAS macro language, and SAS tools for access to non-SAS data, his special interests 
include creation of unique tools to support the SAS BI server and its users, communication-effective visual 
communication and reporting, web information delivery, highly formatted Excel reporting, SAS/GRAPH, ODS, 
and Software-Intelligent Application Development for Reliability, Reusability, Extendibility, and Maintainability. 
He is a regular contributor to VIEWS News, the web newsletter of the VIEWS International SAS Programmer 
Community. 
 
SAS, SAS Enterprise Guide, and all other SAS vendor Inc. product or service names are registered trademarks or trademarks of SAS vendor Inc. 
in the USA and other countries. ® indicates USA registration. Strong Smart Systems and Visual Communication Power are trademarks of LeRoy 
Bessler PhD. Other brand and product names are registered trademarks or trademarks of their respective companies. 
 

Appendix 1.  
 
CPUmon Code To Detect Excessive CPU Usage and Sent Alert Emails 
 
options nosource nonotes nomprint nomprintnest nosymbolgen nomlogic; 
* Turn on one or more above options only for debugging. *; 
* Since monitoring can run for days, weeks, or months, 
  depending on how often the server is rebooted, 
  the SAS log can get very large. *; 
* options source; 
* options notes; 
* options mprint mprintnest; 
* options symbolgen; 
* options mlogic; 
options linesize=max pagesize=max; 
 
*****************************************************************************; 
* Author: LeRoy Bessler PhD                                                 *; 
* Date:   28 February 2010                                                  *; 
* Purpose: Monitor CPU usage of server (or standalone PC) by SAS processes. *; 
* Functions and Options:                                                    *; 
* Monitoring uses ImageName, UserName (User ID), and Process ID (PID)       *; 
* that are delivered in the response to a DOS TaskList command.             *; 
* It uses filter "imagename EQ sas.exe".                                    *; 
* (It could be adapted to monitor any other imagename instead, or as well.) *; 
* Monitor events are separated by time interval WaitSeconds.                *; 
* Optionally it limits the NumberOfMonitorEvents. Macro default is NoLimit. *; 
* It sends an alert if it finds a SAS process that has consumed more than   *; 
* the CPUthresholdInSeconds.                                                *; 
* Each SAS process is identifiable by its PID, or Process ID.               *; 
* The UserID, PID, & CPU time are written to a reference table for alerts.  *; 
* Whenever such a process is detected,                                      *; 
* the reference table is checked for a pre-existing record.                 *; 
* If no record found, the alert is sent, and a reference record is created. *; 
* If there is a pre-existing reference record, and the current CPU time     *; 
* exceeds the previously logged cumulative CPU time by at least             *; 

* the CPUincrementForAnotherAlert, then another alert is sent,              *; 
* and the new CPU time is recorded in that row.                             *; 
* (The CPUincrementForAnotherAlert must be assigned in seconds.)            *; 
* Any PIDs that are no longer active are deleted from the reference table.  *; 
* Alerts are sent directly to the user, with CC to Sender.                  *; 
* If no email address is available for the user,                            *; 
* the alert is sent to the DefaultEmailAddress.                             *; 
* Optionally, every alert can also be sent to the AlwaysNotify address.     *; 
* Email addresses for users are assumed to be in a SAS data set, where      *; 
* User ID is up to 17 characters and Email Address is up to 80 characters.  *; 
* The LookUp table could be different.                                      *; 
* The SAS format built here at monitor start-up could have been pre-built,  *; 
* and OPTIONS FMTSEARCH could be used to point to it.                       *;  
*****************************************************************************; 
 
%macro CPUmon_macro_ForSGFdemo(WaitSeconds=360, 
  NumberOfMonitorEvents=NoLimit, 
  MonLib=, 
  ReportTimeUnits=Hours, /* these units are used for CPU times in the emails */ 
  CPUthresholdInSeconds=7200, 
  CPUincrementForAnotherAlert=3600, 
  PathToEmailAddressDataLib=C:\LeRBmonitors\EmailAddressLib, 
  EmailAddressData=CPUusers, 
  DefaultEmailAddress=Le_Roy_Bessler@wi.rr.com, 
  DefaultExplain28Characters=UserID without Email Address, 
  Sender=Le_Roy_Bessler@wi.rr.com, 
  AlwaysNotify=); 
 
libname EaddrLib "&PathToEmailAddressDataLib"; 
 
data ToFormat; 
length Start $ 17 Label $ 80 HLO $ 1; 
retain fmtname 'Eaddr' type 'C'; 
set EaddrLib.&EmailAddressData(rename=(UserID=Start)) end=LastOne; 
Label = EmailAddress; 
substr(Label,41,40) = FormattedUserName; 
output; 
if LastOne then do; 
  HLO = 'O'; 
  Start = ' '; 
  Label = "&DefaultEmailAddress"; 
  substr(Label,41,40) = "&DefaultExplain28Characters"; 
  output;  
end; 
run; 
 
libname EaddrLib clear; 
 
proc format lib=work CntlIn=ToFormat; 
run; 
 
%let MonitorCycleCount = 0; 
 
libname MonLib "&MonLib"; 
 
%StartOfCycle: 
 
data _null_; 
DateTimeOfMon = datetime(); 
DateOfMon = datepart(DateTimeOfMon); 
TimeOfMon = timepart(DateTimeOfMon); 
call symput('MonDateTime','D'|| trim(left(put(DateOfMon,yymmddn8.))) || '_' || 
     'T'|| trim(left(put(input(compress(put(TimeOfMon,time8.),':'),6.),Z6.))));   
run; 

data _null_; 
length TaskList $ 256; 
TaskList = "'tasklist /v /fi " || '"imagename EQ sas.exe"' || "'"; 
call symput('TaskListCommand',trim(left(TaskList))); 
run; 
 
filename tasklist pipe &TaskListCommand; 
 
data 
MonLib.ActiveProcesses 
(keep=UserID Process_ID     Cumulative_CPU_seconds     MonDT OrigID)   
%if &MonitorCycleCount eq 0 
%then %do; 
MonLib.CPUmonRefTable 
(keep=UserID Process_ID Ref_Cumulative_CPU_seconds Ref_MonDT) 
%end; 
                      ; 
length OrigID UserID $ 17 Process_ID $ 6 Cumulative_CPU_seconds 8 MonDT $ 21; 
infile tasklist lrecl=229 pad; 
input @1 line $char229.; 
if _N_ GE 4; 
OrigID = substr(substr(line,89,50),17,17); 
UserID = upcase(OrigID); 
Process_ID = left(substr(line,27,6)); 
Cumulative_CPU_seconds = input(trim(left(substr(line,140,12))),hhmmss12.); 
MonDT = put(datetime(),datetime21.2); 
drop line; 
 /* Site-customize and UnComment the section BELOW for testing, if desired. 
file  
%if &MonitorCycleCount eq 0 
%then %do; 
     "C:\LeRBmonitors\TasksFoundAtStartUp_&ReportTimeUnits..txt" 
%end; 
%else %do; 
     "C:\LeRBmonitors\TasksFoundThisCycle_&ReportTimeUnits..txt" 
%end; 
                                               lrecl=229; 
put @1 line $char229.; 
    Site-customize and UnComment the section ABOVE for testing, if desired. */ 
output MonLib.ActiveProcesses; 
%if &MonitorCycleCount eq 0 
%then %do; 
Ref_Cumulative_CPU_seconds = Cumulative_CPU_seconds; 
Ref_MonDT                  = MonDT; 
output MonLib.CPUmonRefTable; 
%end; 
run; 
   
proc sort data=MonLib.CPUmonRefTable; 
by Process_ID UserID; 
run; 
   
proc sort data=MonLib.ActiveProcesses; 
by Process_ID UserID; 
run; 
 
data 
Alerts               (keep=OrigID Process_ID     Cumulative_CPU_seconds     MonDT) 
MonLib.CPUmonRefTable(keep=UserID Process_ID Ref_Cumulative_CPU_seconds Ref_MonDT); 
merge MonLib.CPUmonRefTable(in=Ref) MonLib.ActiveProcesses(in=Active); 
by Process_ID UserID; 
if Active; 

if Cumulative_CPU_Seconds GT &CPUthresholdInSeconds; 
      if NOT Ref 
      then do; 
        output Alerts; 
        Ref_Cumulative_CPU_seconds = Cumulative_CPU_seconds; 
        Ref_MonDT                  = MonDT; 
      end; 
      else do; 
%if &MonitorCycleCount eq 0 
%then %do; 
        output Alerts;  
%end; 
%else %do; 
        if Cumulative_CPU_seconds GE  
           Ref_Cumulative_CPU_seconds + &CPUincrementForAnotherAlert 
        then do; 
          output Alerts; 
          Ref_Cumulative_CPU_seconds = Cumulative_CPU_seconds; 
          Ref_MonDT                  = MonDT;          
        end; 
%end; 
      end; 
output MonLib.CPUmonRefTable; 
run; 
 
data _null_; 
call symput('AlertsCount',ObsCount); stop; 
set Alerts nobs=ObsCount; 
run; 
 
%if %eval(&AlertsCount GT 0) 
%then %do; 
data _null_; 
length AddressAndName Address Name $ 80; 
set Alerts; 
AddressAndName = put(OrigID,$Eaddr.); 
Address = substr(AddressAndName,01,40); 
Name    = substr(AddressAndName,41,40); 
if Name EQ "&DefaultExplain28Characters" 
then Name = trim(left(OrigID)) || " is &DefaultExplain28Characters";  
call symput('MonDT'||trim(left(_N_)),trim(left(MonDT))); 
call symput('ID'   ||trim(left(_N_)),trim(left(OrigID))); 
call symput('Addr' ||trim(left(_N_)),trim(left(Address))); 
call symput('Name' ||trim(left(_N_)),trim(left(Name))); 
call symput('PID'  ||trim(left(_N_)),trim(left(Process_ID))); 
  %if %upcase(&ReportTimeUnits) = HOURS 
  %then %do; 
call symput('Hrs'  ||trim(left(_N_)), 
            trim(left(put(Cumulative_CPU_seconds      /3600,6.2))));   
call symput('MsgIncrementValue'     , 
            trim(left(put(&CPUincrementForAnotherAlert/3600,6.2)))); 
call symput('MsgTriggerValue'       , 
            trim(left(put(&CPUthresholdInSeconds      /3600,6.2)))); 
  %end; 
  %else 
  %if %upcase(&ReportTimeUnits) = MINUTES %then %do; 
call symput('Mins' ||trim(left(_N_)), 
            trim(left(put(Cumulative_CPU_seconds      /60,comma8.1)))); 
call symput('MsgIncrementValue'     , 
            trim(left(put(&CPUincrementForAnotherAlert/60,comma8.1)))); 
call symput('MsgTriggerValue'       , 
            trim(left(put(&CPUthresholdInSeconds      /60,comma8.1)))); 
  %end; 

  %else %do; 
call symput('Secs' ||trim(left(_N_)),trim(left(Cumulative_CPU_seconds      ))); 
call symput('MsgIncrementValue'     ,trim(left(&CPUincrementForAnotherAlert))); 
call symput('MsgTriggerValue'       ,trim(left(&CPUthresholdInSecond       ))); 
  %end; 
run; 
 
  %do i = 1 %to &AlertsCount %by 1; 
 
    %if &&PID&i NE &sysjobid 
    %then %do;  
 
%if %upcase(&ReportTimeUnits) = HOURS 
%then %let ReportTimeValue = &&Hrs&i; 
%else 
%if %upcase(&ReportTimeUnits) = MINUTES 
%then %let ReportTimeValue = &&Mins&i; 
%else %let ReportTimeValue = &&Secs&i; 
%let Msg = 
User ID &&ID&i Process &&PID&i Has Used &ReportTimeValue &ReportTimeUnits of CPU 
Time.; 
%let IncrementMsg = 
You will get a message every time process &&PID&i uses another &MsgIncrementValue 
&ReportTimeUnits of CPU time.; 
%let TriggerMsg = 
or it has used at least &MsgTriggerValue &ReportTimeUnits of CPU time and you get this 
message.; 
 
filename AnyEmail EMAIL 
FROM="&Sender" 
SENDER="&Sender" 
TO=("&&Addr&i") 
CC=("&Sender" 
%if %length(&AlwaysNotify) NE 0 %then %do; 
    "&AlwaysNotify" 
%end; 
) 
SUBJECT="&Msg Is it OK?"; 
 
data _null_; 
file AnyEmail; 
put "Dear &&Name&i"; 
put ' '; 
put "Please read this long automated message carefully and in entirety."; 
put ' '; 
put "As Of &&MonDT&i"; 
put "&Msg"; 
put "Is it OK? If not, please terminate it."; 
put ' '; 
put "For information on how to terminate a SAS process, go to"; 
put "the URL for that part of the SAS Server Site User Documentation."; 
put "If you can not terminate it, call the SAS Server Administrator."; 
put "If the SAS Server Administrator is unavailable, call the Help Desk."; 
put "They need the Process ID &&PID&i and Server Name %sysget(computername)."; 
put ' '; 
put "&IncrementMsg"; 
put "The process ID &&PID&i was assigned when your SAS Enterprise Guide or Display 
Manager session or batch job started."; 
put "It varies from session to session and from job to job."; 
put ' '; 
put "CPU time is NOT elapsed time. It is active processing time."; 
put "The CPU waits idle for file reads, file writes, responses from Oracle, etc."; 
put "CPU time is usually less than elapsed time."; 

put ' '; 
put "Using a lot of CPU time is not necessarily bad."; 
put "Heavy-duty analytics processes can use a lot of CPU time."; 
put "However, if your process is in a loop, it is wasting resources."; 
put "If your process is from an abandoned or frozen Enterprise Guide session,"; 
put "you will be unaware of how much CPU time it is using"; 
put "unless you investigate with the process management tools,"; 
put "&TriggerMsg"; 
put "If you have any questions, call or send email to the SAS Server Administrator."; 
run; 
 
    %end; 
 
  %end; 
 
%end; 
 
%let MonitorCycleCount = %eval(&MonitorCycleCount + 1); 
 
%put This monitor cycle &MonitorCycleCount ran at &MonDateTime; 
 /* If there is an ERROR or WARNING message in the log, 
    the above statement lets you estimate the time of that message, 
    which is NOT datetimestamped by SAS software. */ 
  
%if %upcase(&NumberOfMonitorEvents) NE NOLIMIT %then %do; 
  %if %eval(&MonitorCycleCount GE &NumberOfMonitorEvents) 
  %then %GoTo EndOfMonitorSession; 
%end; 
 
data _null_; 
x = sleep(&WaitSeconds);  
run;                
 
%GoTo StartOfCycle; 
 
%EndOfMonitorSession: 
 
%mend CPUmon_macro_ForSGFdemo; 
 
%CPUmon_macro_ForSGFdemo(WaitSeconds=360, 
              NumberOfMonitorEvents=NoLimit, 
              MonLib=PathToFolderForOutputCPUmonLogDataSets, 
              ReportTimeUnits=Hours, 
              CPUthresholdInSeconds=7200, 
              CPUincrementForAnotherAlert=3600, 
              PathToEmailAddressDataLib=C:\LeRBmonitors\EmailAddressLib, 
              EmailAddressData=CPUusers, 
              DefaultEmailAddress=Le_Roy_Bessler@wi.rr.com, 
              DefaultExplain28Characters=UserID without Email Address, 
              Sender=Le_Roy_Bessler@wi.rr.com, 
              AlwaysNotify=LeRB.BackUp@somewhere.com); /* using mostly the defaults */ 
 
LogTimer Macro 
 
%MACRO LogTimer(StartOrEnd); 
 
%LET StartOrEnd = %UPCASE(&StartOrEnd.); 
%GLOBAL LogTimerWasStarted; 
%GLOBAL SaveStartDateTime; 
%GLOBAL SaveStartCPUtime; 
 
DATA _NULL_; 
DateTime = DATETIME(); 

CALL SYMPUT('LogTimerDate',TRIM(LEFT(PUT(DATEPART(DateTime), DATE9.)))); 
CALL SYMPUT('LogTimerTime',TRIM(LEFT(PUT(TIMEPART(DateTime), TIME8.)))); 
IF "&StartorEnd" = 'START' 
THEN CALL SYMPUT('SaveStartDateTime',PUT(DateTime, 22.10)); 
 /* maximum precision for saved Start datetime 
    to avoid possible negative elapsed time */ 
ELSE DO; 
  ElapsedSeconds = DateTime - INPUT(SYMGET('SaveStartDateTime'), 22.10); 
  CALL SYMPUT('LogTimerElapsedTime',TRIM(LEFT(PUT(ElapsedSeconds, TIME.)))); 
END; 
RUN; 
 
%LET TaskListCommand = %STR('tasklist /v'); 
FILENAME TaskList PIPE &TaskListCommand.; 
DATA _NULL_; 
INFILE TaskList LRECL = 224 PAD END = LastOne; 
 /* LRECL varies by Windows version. 
    224 is appropriate for Windows XP. 
    229 is for Windows 2003 Advanced Server. */ 
INPUT @1 CommandResponse $CHAR224.; 
IF _N_ GE 4; 
IF INDEX(CommandResponse, "&sysjobID.") NE 0; 
Cumulative_CPUtime = TRIM(LEFT(SUBSTR(CommandResponse, 140, 12))); 
 /* offset of Cumulative_CPU_time varies by Windows version. 
    140 is appropriate for Windows XP. 
    145 is for Windows 2003 Advanced Server. */ 
Cumulative_CPU_seconds = INPUT(Cumulative_CPUtime, HHMMSS12.); 
IF "&StartorEnd." = 'START' 
THEN CALL SYMPUT('SaveStartCPUtime',TRIM(LEFT(Cumulative_CPU_seconds))); 
ELSE DO; 
  CPU_seconds = Cumulative_CPU_seconds - INPUT(SYMGET('SaveStartCPUtime'), 8.); 
  CALL SYMPUT('LogTimerCPUtime',TRIM(LEFT(PUT(CPU_seconds, TIME.)))); 
END; 
RUN; 
%PUT *****************************************; 
%IF &StartOrEnd. = START 
%THEN %DO; 
  %PUT Started at &LogTimerTime. On &LogTimerDate.; 
  %LET LogTimerWasStarted = YES; 
%END; 
%ELSE 
%IF &StartOrEnd. = END 
%THEN %DO; 
  %IF &LogTimerWasStarted. EQ YES 
  %THEN %DO; 
    %PUT Ended at &LogTimerTime. on &LogTimerDate.; 
    %PUT Elapsed Time (hours:minutes:seconds) = &LogTimerElapsedTime.; 
    %PUT CPU Time (hours:minutes:seconds) = &LogTimerCPUtime.; 
  %END; 
  %ELSE %DO; 
    %PUT LogTimer Macro User ERROR: Invocation Value was &StartOrEnd.; 
    %PUT But there was no prior invocation with Start; 
  %END; 
%END; 
%ELSE %DO; 
  %PUT LogTimer Macro User ERROR: Invocation Value was &StartOrEnd.; 
  %PUT Must be Start or End; 
%END; 
%PUT *****************************************; 
 
%MEND LogTimer; 

UserMon Code To Create UserMon Logs 
 
options nosource nonotes nomprint nomprintnest nosymbolgen nomlogic; 
/* Turn on one or more above options only for debugging. Since monitoring can run for 
   days, weeks, or months, the SAS log can get very large. */ 
 
options pagesize=max; 
 
%macro UserMon(WaitSeconds=360,NumberOfMonitorEvents=NoLimit,MonLib=);  
 
%let MonitorCycleCount = 0; 
%let DateOfLog = 0; 
%let TimeOfLog = 0; 
 
libname MonLib "&MonLib"; 
 
%StartOfCycle: 
 
data _null_; 
DateOfLog = datepart(datetime()); 
TimeOfLog = timepart(datetime()); 
if DateOfLog GT &DateOfLog 
  or 
   (DateOfLog EQ &DateOfLog and TimeOfLog GT &TimeOfLog) 
then do;     
  call symput('DateOfLog',trim(left(DateOfLog))); 
  call symput('TimeOfLog',trim(left(TimeOfLog))); 
  call symput('LogDateTime','D'|| trim(left(put(DateOfLog,yymmddn8.))) || '_' || 
       'T'|| trim(left(put(input(compress(put(TimeOfLog,time8.),':'),6.),Z6.)))); 
end;   
run; 
 
data _null_; 
length TaskList $ 256; 
TaskList = "'tasklist /v /fi " || '"imagename EQ sas.exe"' || "'"; 
call symput('TaskListCommand',trim(left(TaskList))); 
run; 
 
filename tasklist pipe &TaskListCommand; 
    
data MonLib.UserMonLog_&LogDateTime(drop=line Memory_Usage); 
length MonDateTime $ 21 UserName $ 50 Process_ID $ 6 Cumulative_CPUtime $ 12 
       Cumulative_CPU_seconds Current_Memory_Usage_In_KB MonDT 8; 
retain MonDateTime '01JAN1960:00:00:01.00' MonDt 0; 
infile tasklist lrecl=229 pad; 
input @1 line $char229.; 
if _N_ EQ 1 then do; 
  MonDT = datetime(); 
  MonDateTime = put(MonDT,datetime21.2); 
  call symput('MonDateTime',trim(left(MonDateTime))); 
end; 
if _N_ GE 4; 
Process_ID   = left(substr(line,27,8)); 
Memory_Usage = substr(line,65,12);  
Current_Memory_Usage_In_KB = 
  input(substr(Memory_Usage,1,index(Memory_Usage,'K') - 2),comma10.); 
UserName     = substr(line,94,50); 
Cumulative_CPUtime = trim(left(substr(line,145,12))); 
Cumulative_CPU_seconds = input(Cumulative_CPUtime,hhmmss12.); 
run; 
 
%let MonitorCycleCount = %eval(&MonitorCycleCount + 1); 
%put This monitor cycle &MonitorCycleCount ran at &MonDateTime; 

* If there is an ERROR or WARNING message in the log, the above statement lets you 
estimate the time of that message, which is NOT datetimestamped by SAS software. *; 
  
%if %upcase(&NumberOfMonitorEvents) NE NOLIMIT %then %do; 
  %if %eval(&MonitorCycleCount EQ &NumberOfMonitorEvents) 
  %then %GoTo EndOfMonitorSession; 
%end; 
 
data _null_; 
x = sleep(&WaitSeconds);  
run;                
 
%GoTo StartOfCycle; 
 
%EndOfMonitorSession: 
 
%mend UserMon; 
 
%UserMon(MonLib=PathToFolderForOutputUserMonLogDataSets); 
 
ConcatMonLogs Macro and Its Invocation Code To Concatenate UserMon Logs 
 
%macro ConcatMonLogs(MonLib=,OutLib=SASUSER,out=ConcatMonLogs, 
StartYYYYMMDD=00000000,EndYYYYMMDD=99999999,OmitLastMonLog=YES); 
 
/* OmitLastMonLog=YES captures logs through the second latest one in MonLib */ 
/* This prevents conflict if EndYYYYMMDD is left at macro default, 
   but UserMon is still writing the latest Monitor Log in MonLib. */ 
 
%global StartDT EndDT; /* for reference by subsequent processing outside of macro */  
 
libname MonLogs "&MonLib"; 
 
proc sql; 
create table work.MonLogs as 
(select libname, memname from dictionary.tables where libname EQ 'MONLOGS'); 
quit; 
 
proc sort data=work.MonLogs; by memname; run; 
 
data work.MonLogs; 
set work.MonLogs(where=("&StartYYYYMMDD" LE substr(memname,12,8) LE "&EndYYYYMMDD")); 
run; 
 
data _null_; 
call symput('HowMany',trim(left(HowMany))); 
stop; 
set work.MonLogs nobs=HowMany; 
run; 
 
%if &HowMany EQ 0 %then %do; 
  %put No Logs in Date Range &StartYYYYMMDD through &EndYYYYMMDD; 
  %goto MacExit; 
%end; 
 
data _null_; 
set work.MonLogs end=LastOne; 
if _N_ EQ 1 then call symput('StartDT',substr(memname,11,17)); 
if not LastOne then do; 
  call symput('Log'||trim(left(_N_)),trim(left(memname))); 
  call symput('EndDT',substr(memname,11,17)); 
end; 
else do; 

  flag = "&OmitLastMonLog"; 
  if upcase(flag) EQ 'YES' 
  then call symput('LogCount',_N_ - 1); 
  else do; 
    call symput('Log'||trim(left(_N_)),trim(left(memname))); 
    call symput('EndDT',substr(memname,11,17)); 
    call symput('LogCount',_N_); 
  end;  
end; 
run; 
 
proc datasets lib=&OutLib nodetails nolist; 
delete &out; run; quit; 
 
%do i = 1 %to &LogCount %by 1; 
proc append base=&OutLib..&out data=MonLogs.&&Log&i force; run;  
%end; 
 
%MacExit: 
 
libname MonLogs clear; 
 
%mend  ConcatMonLogs; 
 
%ConcatMonLogs(MonLib= PathToFolderForUserMonLogDataSets, 
out=ConcatMonLogsJan2009,OmitLastMonLog=NO, 
StartYYYYMMDD=20090101,EndYYYYMMDD=20090131); 
 
Example of a Query Macro to Analyze Monitor Logs 
 
%macro WorkloadSummaryHistory(data=,RptPath=); 
 
proc summary data=&data nway; 
id MonDT; 
class MonDateTime; 
var Current_Memory_Usage_In_KB; 
output out=ByMonitorInterval(rename=(_freq_=Number_of_Users_Or_Jobs)) 
sum=Total_Memory_Usage_In_KB; 
run; 
 
data ByMonitorInterval(drop=Day_Of_Week_Number); 
length Day_Of_Week $ 9 Month $ 3 Day_Of_Month $ 2 Hour $ 2; 
set ByMonitorInterval; 
Hour = put(hour(MonDT),Z2.); 
sasDate = datepart(MonDT); 
Month = put(sasDate,monname3.); 
Day_Of_Month = day(sasDate);  
Day_Of_Week_Number = weekday(datepart(MonDT)); 
if Day_Of_Week_Number EQ 1 
then Day_Of_Week = 'Sun'; 
else 
if Day_Of_Week_Number EQ 2 
then Day_Of_Week = 'Mon'; 
else 
if Day_Of_Week_Number EQ 3 
then Day_Of_Week = 'Tue'; 
else 
if Day_Of_Week_Number EQ 4 
then Day_Of_Week = 'Wed'; 
else 
if Day_Of_Week_Number EQ 5 
then Day_Of_Week = 'Thu'; 
else 

if Day_Of_Week_Number EQ 6 
then Day_Of_Week = 'Fri'; 
else Day_Of_Week = 'Sat'; 
run; 
 
proc summary data=ByMonitorInterval nway; 
id Day_Of_Week Month Day_Of_Month; 
class sasDate Hour; 
var Number_of_Users_Or_Jobs Total_Memory_Usage_In_KB; 
output out=ByHourWithinDay(drop=_freq_) max=Max_Users_Or_Jobs 
Max_Total_Memory_Usage_In_KB; 
run; 
 
 /* TITLE2 retrieves StartDT and EndDT passed by ConcatMonLogs macro */ 
 
title1 "Server SAS WorkLoad By Hour"; 
title2 "From &StartDT To &EndDT"; 
title3 "NOTE: Processes shorter than 6 minutes can be missed"; 
title1; /* If you turn on the titles, use HTML column spanning to prevent Column A 
           being stretched to the width of the longest title line. */ 
 
ods html file="&RptPath.\WorkloadByHourWithinDay.xls" style=Minimal; 
proc print data=ByHourWithinDay split='_' noobs; 
var Day_Of_Month Month Day_Of_Week Hour Max_Users_Or_Jobs 
Max_Total_Memory_Usage_In_KB;  
run; 
ods html close; 
 
proc sort data=ByMonitorInterval; 
by MonDT; 
run; 
 
title1 "Server SAS WorkLoad By Monitor Interval"; 
title2 "From &StartDT To &EndDT"; 
title3 "NOTE: Processes shorter than 6 minutes can be missed"; 
title1; /* If you turn on the titles, use HTML column spanning to prevent Column A 
           being stretched to the width of the longest title line. */ 
 
ods html file="&RptPath.\WorkloadByMonitorInterval.xls" style=Minimal; 
proc print data=ByMonitorInterval split='_' noobs; 
* where '25Jan2009'd LE datepart(MonDT) LE '25Jan2009'd;  
var MonDateTime Day_Of_Week Number_of_Users_Or_Jobs Total_Memory_Usage_In_KB; 
label MonDateTime='Monitor DateTime';  
run; 
ods html close; 
 
%mend WorkloadSummaryHistory; 
 
%WorkloadSummaryHistory(data=SASUSER.ConcatMonLogsJan2009,RptPath=PathToReports); 
 
ShowProcessID Macro 
 
/* Run this macro at the start of your SAS Enterprise Guide session to be able to 
distinguish your current SAS Enterprise Guide session’s process from the other one(s) 
that you want to display and possibly terminate. */ 
 
%macro ShowProcessID; 
 
%put Process ID for this SAS Enterprise Guide session or SAS batch job is &sysjobid; 
 
%mend  ShowProcessID; 

DisplayAllMySASprocesses Macro 
 
%macro DisplayAllMyProcesses; 
 
/* To provide a compact listing of the process information, 
   the TaskList command response must be parsed. 
   The layout of the response varies by Windows version. 
   The offsets used below are for Windows XP Professional Version 2002. */ 
 
data _null_; 
length cmd $ 100; 
cmd = "'tasklist /v" || "'"; 
call symput('taskcmd',trim(left(cmd))); 
run; 
 
filename TaskList pipe &taskcmd; 
 
data _null_; 
length PID $ 6 Memory $ 12 CPUtime $ 12; 
infile TaskList lrecl=229 pad end=LastOne; 
input @1 CommandResponse $char229.; 
if _N_ EQ 1 then do; 
  put @1 "====== ============ ============"; 
  put @1 "PID    Memory Usage CPU Time";   
  put @1 "====== ============ ============"; 
end; 
if _N_ GE 4 
  and 
   index(CommandResponse,"&sysuserID") NE 0 
  and 
   CommandResponse =: 'sas.exe' 
then do; 
  PID     = left(substr(CommandResponse,25,8)); 
  Memory  = substr(CommandResponse,60,12);  
  CPUtime = substr(CommandResponse,140,12); 
  put 
  @1  PID      $6. 
  @8  Memory  $12. 
  @21 CPUtime $12.;   
end; 
if LastOne 
then put @1 "====== ============ ============";  
run; 
 
%mend  DisplayAllMyProcesses; 
 
%macro DisplayAllMySASprocessesUnParsed; 
 
/* To provide a compact listing of the process information, 
   the TaskList command response must be parsed. 
   The layout of the response varies by Windows version. 
   This macro can be used during development. */ 
 
%global SaveLineSize;  
%let SaveLineSize = %sysfunc(getoption(LineSize)); 
options LineSize=229; 
filename TaskList pipe 'tasklist /v'; 
data _null_; 
infile TaskList lrecl=229 pad; 
input @1 CommandResponse $char229.; 
if _N_ LT 4 then put CommandResponse; 
else 
if index(CommandResponse,"&sysuserID") NE 0 

  and 
   CommandResponse =: 'sas.exe' 
then put CommandResponse;    
run; 
options LineSize=&SaveLineSize; 
 
%mend  DisplayAllMySASprocessesUnParsed; 
 

TerminateProcess Macro 
 
%macro TerminateProcess(ProcessID=); 
 
data _null_; 
cmd = "'tasklist /v /fi " ||'"' || "PID EQ &ProcessID" || '"' || "'"; 
call symput('taskcmd',trim(left(cmd))); 
run; 
filename TaskList pipe &taskcmd; 
data _null_; 
infile TaskList lrecl=229 pad; 
input @1 CommandResponse $char229.; 
if CommandResponse EQ 'INFO: No tasks are running which match the specified criteria.' 
then do; 
  call symput('Found','N'); 
  put "Process ID &ProcessID was not found."; 
end; 
else do; 
  if _N_ GE 4; 
  if index(CommandResponse,"&sysuserid") EQ 0 then do; 
    call symput('Found','N'); 
    put "Process ID &ProcessID is not for User ID &sysuserid and will not be killed."; 
  end; 
  else call symput('Found','Y'); 
end; 
run; 
 
%if &Found EQ N %then %goto MacExit; 
 
data _null_; 
cmd = "'taskkill /fi " ||'"' || "PID EQ &ProcessID" || '"' || "'"; 
call symput('taskcmd',trim(left(cmd))); 
run; 
filename TaskKill pipe &taskcmd; 
data _null_; 
infile TaskKill lrecl=229 pad; 
input @1 CommandResponse $char229.; 
put CommandResponse; 
run; 
 
%MacExit: 
%mend  TerminateProcess;

Chapter 3: Local Web-Based SAS User Documentation and Tools Environment 
 
What I developed for my former employer was a derivative and site-specific enhancement of a 
package that I originally developed so that attendees at our Wisconsin Illinois SAS Users 
Conferences could go back to their offices with much, much more than just a concatenation of 
conference papers and slides. Rather than disclose a corporate proprietary internal environment, 
this chapter will show you our users conference Proceedings and Tools web-enabled CD content. 
Besides a collection of folders of various information, the CD always includes the SAS code that 
was required to create its web-enabled front-end. It can be loaded to a stand-alone PC, to a 
server, or to any LAN folder and then used as SAS User Documentation. The provided SAS code 
can be modified however desired to create an altered and/or enhanced alternate environment. 
 
First, let me present screen images of some of the CD’s web pages. 
 
Contents of Folder Loaded from a CD or a from an Emailed Zip File of the CD 

Home Page (After Clicking StartHere.html) 

Structure of the Home Page (and Overview of the Deliverables) 
 
Local Resources 
 
About the Speakers, Other Contributing Authors, and the Volunteers 
Conference Slides 
Conference Papers 
Extra Slides, Papers, and Reference Documents 
Sample SAS Programs 
 
These are Conference-specific resources provided in folders on the CD (or in the zip file). For a 
SAS user site, these could be replaced with whatever is deemed useful. 
 
Remote Resources 
 
Ways/Places To Get SAS Institute Help and Information 
SAS Information at Other Remote Sources 
Wisconsin Illinois SAS Users Home Page 
 
The last item is users-group specific. One instead could link to any other users group(s)—
internal, local, subregional, regional, international, special interest (e.g., PharmaSUG). 
 
At Assurant Health, rather than only one link to the avalanche of information at support.sas.com, 
I provided guided links to segments of that environment, both of general interest and pertinent to 
the company’s SAS software portfolio and tools.  

After clicking Ways/Places To Get SAS Institute Help and Information 

After clicking SAS Information at Other Remote Sources 

Code Used to Create the Web Page Above (excerpted from the full program) 
 
ods html path="&Path" 
         (url=none) 
         body='RemoteTips&News&Links.html' 
         (title='SAS Information at Other Remote Sources') 
         style=styles.LeRBrecommended; 
 
title1 j=L 
c=CX000000 h=&LargeFontSize 
'SAS Information at Other Remote Sources'; 
title2 
j=L 
' ' 
j=L 
link='http://www.views-uk.org' 
'"VIEWS, the International SAS Programmer Community": Home Page, How To Join, etc.' 
j=L 
link='http://www.views-uk.demon.co.uk/Newsletter/backcopy.htm' 
'To download back-issues of the VIEWS newsletter' 
j=L 
' ' 
j=L 
link='http://www.lexjansen.com' 
'Explore Lex Jansen''s Links to a Treasure Trove of SAS Lore' 
j=L 
link='http://www.sconsig.com' 
'And Still More Links . . . Including the Time and Temperature Somewhere' 
j=L 
' ' 
j=L 
link='http://www.sasCommunity.org' 
'Wiki-based SAS Users Information Exchange Environment' 
j=L 
' ' 
j=L 
'How To Ask Questions Of, and Provide Answers To,' 
j=L 
'Other SAS Users Around the World' 
j=L 
'At a User-Sponsored Forum For Anything About SAS:' 
j=L 
link='FromCD\Addenda\Kelley - SAS_L - The Discussion That Never Stops.doc' 
'F. Joseph Kelley - "SAS-L: The Discussion That Never Stops"' 
j=L 
link='FromCD\Addenda\Kelley - SAS_L - The Basics.pdf' 
'F. Joseph Kelley - "SAS-L: The Basics"' 
j=L ' ' j=L 
link='StartHere.html' 
'Home Page'; 
 
%WebPubStdFootNote(Year=&YEAR,FtNumber=1, 
                   WebPageBackgroundRGBcolor=&WebPageBackground); 
%HiddenFakeTable(BaseStyle=LeRBrecommended, 
                   WebPageBackgroundRGBcolor=&WebPageBackground); 

The full program is supplied on the CD (or zip file): 
 

  
 

The macros used by the program are supplied on the CD (or zip file): 

Here is the program information header: 
 
**********************************************************; 
* Program: SUS_WebPubPgm_SAScode.sas                     *; 
*          All Rights Reserved. Software User Services.  *; 
*          Fox Point, Wisconsin, USA                     *; 
* Mailing                                                *; 
* Address: PO Box 96, Milwaukee, WI 53201-0096, USA      *; 
* History: First developed in 2005 to support the        *; 
*          Wisconsin Illinois SAS Users Conference.      *; 
*          Revised yearly for conferences.               *; 
* Latest Revision Date: 23 June 2009                     *; 
* Purpose: Provide a web table of contents and links     *; 
*          to ANY collection of electronic resoources,   *; 
*          but designed by default for specific items    *; 
*          that it assumes have been pre-loaded into     *; 
*          pre-allocated folders.                        *; 
* Note0:   This SAS program contains commented out code  *; 
*          that could be used to provide                 *; 
*          a link to local SAS resources at your site.   *; 
* Note1:   This SAS program is distributed               *; 
*          with companion macros used by it, and         *; 
*          with the designed-in folders of resources.    *; 
*          Those designed-in folders can be ignored,     *; 
*          but then all links to them must be deleted    *; 
*          from this program.                            *; 
*          Regardless of what is done                    *; 
*          with the designed-in folders and their links, *; 
*          new folders can be added, and links to them.  *; 
* Note2:   The %CreateCustomStyle macro used here can be *; 
*          invoked with other assignments if you want    *; 
*          to change the appearance of the web pages.    *; 
*          The parameters assigned here are the same as  *; 
*          the macro defaults.                           *; 
* Note3:   The %CreateCustomStyle macro must have been   *; 
*          pre-loaded into &MacroFolder as assigned.     *; 
* Note4:   Use of the %CreateCustomStyle macro is        *; 
*          NOT mandatory, but if not used, you must      *; 
*          revise your use of the %HiddenFakeTable macro *; 
*          as appropriate.                               *; 
* Note5:   %WebPubStdFootNote & %HiddenFakeTable macros  *; 
*          used here must ALSO have been pre-loaded.     *; 
* Note6:   The WebPageBackgroundRGBcolor assigned with   *; 
*          the %HiddenFakeTable macro MUST MATCH the     *; 
*          WebPageBackgroundRGBcolor assigned with the   *; 
*          %CreateCustomODSstyle macro. If you do not    *; 
*          use the %CreateCustomODSstyle macro, then you *; 
*          must determine the web page background of the *; 
*          ODS style that you are using, and MATCH it.   *; 
**********************************************************;

Here is the program initialization code: 
 
%let YEAR = 2009; * CHANGE AS NEEDED *********************; 
 
%let path = C:\SUS&YEAR.CD; 
* Adjust above storage location for web pages as appropriate                   *; 
* Any folders referenced by Path or elsewhere below must be pre-built & loaded *; 
 
%let MacroFolder = %str(&Path.\FromCD\SUSmacros); 
options sasautos=(sasautos "&MacroFolder") mautosource; 
 
%let WebPageBackground = CXFFFF99; 
* light (not lightest) Browser-Safe yellow      *; 
* lightest yellow may wash out on an LED screen *; 
* This color is used by %CreateCustomODSstyle 
                 and by %HiddenFakeTable        *; 
 
%let LargeFontSize  = 16pt; 
 
options mprint; /* show code generated by macro, and executed */ 
 
%CreateCustomODSstyle 
(CustomStyle=LeRBrecommended, 
WebPageBackgroundRGBcolor=&WebPageBackground, 
TableTitleFootnoteFont=Georgia, 
TableTitleFootnoteSize=4, 
TableHeadingFont=Georgia, 
TableHeadingSize=1, 
TableDataFont=Verdana, 
TableDataSize=1, 
TableFrame=void, /* use TableFrame=void to remove frame */ 
TableFrameRGBcolor=CX9999FF, /* light (not lightest) Browser-Safe blue */ 
TableGrid=NO, /* use YES to turn on grid between data cells */ 
TableSpacing=0); /* the SAS-shipped default is 7, 
                    space between cell data and cell boundary */ 
 
ods noresults; 
ods listing close; 

Below is one of two key macros used by the program. See the explanation of its Purpose in the 
macro header text. As of Version 9.1.3, the comment about SAS/GRAPH is obsolete. 
 
%macro HiddenFakeTable 
       (BaseStyle=, 
        WebPageBackgroundRGBcolor=CXFFFF99); 
 
**********************************************************; 
* Author:  LeRoy Bessler PhD - bessler@execpc.com        *; 
* Date:    11 June 2005                                  *; 
* Purpose: Create an empty table                         *; 
*          that is used in an ODS trick                  *; 
*          to be able to create hyperlinks               *; 
*          on a web page                                 *; 
*          with the LINK= option                         *; 
*          on table TITLEn and/or FOOTNOTEn statements.  *; 
*          Useful for an output-free web page,           *; 
*          or a graph-only web page.                     *; 
*          SAS/GRAPH does not support the LINK= option.  *; 
* Usage:   You must assure that the table colors         *; 
*          match the web page background color.          *; 
*          Find a sample use of this macro by LeRB.      *; 
*          Use at your own risk.                         *; 
*          There is an edit                              *; 
*          of your macro invocation parameters,          *; 
*          but that is no guarantee that                 *; 
*          it cannot be improperly invoked               *; 
*          with adverse consequences.                    *; 
**********************************************************; 
 
%if %length(&BaseStyle) eq 0 
%then %do; 
  %put MACRO USER ERROR: BaseStyle was not assigned; 
  %goto MacroExit; 
%end; 
 
%if %length(&WebPageBackgroundRGBcolor) eq 0 
%then %do; 
  %put MACRO USER ERROR: WebPageBackgroundRGBcolor was not assigned; 
  %goto MacroExit; 
%end; 
 
%global PrepForHiddenTableComplete; 
 
%if %length(&PrepForHiddenTableComplete) eq 0 %then %do; 
 
data FakeData; 
length X $ 1; 
X = 'X'; 
run; 
 
%let PrepForHiddenTableComplete = YES; 
 
%end; 
 
proc print data=FakeData noobs 
     style=[rules=none frame=void cellspacing=0 cellpadding=0 borderwidth=0]  
     style(header)=[foreground=&WebPageBackgroundRGBcolor 
                    background=&WebPageBackgroundRGBcolor 
                    font_size=1] 
     style(data)=[foreground=&WebPageBackgroundRGBcolor 
                    background=&WebPageBackgroundRGBcolor 

                    font_size=1]; 
run; 
 
ods html close; 
 
%MacroExit: 
 
%mend  HiddenFakeTable; 

Below is another of two key macros used by the program. It demonstrates the counterintuitive, 
important, but probably realized by most users, point that use of line breaks in a FOOTNOTE 
statement requires defining the desired text lines in reverse order. I.e., segments of the multi-line 
footnote that are top to bottom in the FOOTNOTE statement end up bottom to top in the 
displayed result. 
 
%macro WebPubStdFootNote(FtNumber=1,WebPageBackgroundRGBcolor=CXFFFF99,Year=2008, 
                         SAS8defaultFontSize=4pt,SAS9defaultFontSize=8pt); 
options nosource; 
 
**********************************************************; 
* Macro:   WebPubStdFootNote                             *; 
* Author:  LeRoy Bessler PhD - bessler@execpc.com        *; 
* Date:    17 June 2005                                  *; 
* Revised: 4 May 2008                                    *; 
* Purpose: Create a standard footnote                    *; 
*          for the WebPub pages.                         *; 
* Usage:   If other footnotes are being used,            *; 
*          override FtNumber=1 as appropriate.           *; 
*          Maximum value is 10.                          *; 
*          Find a sample use of this macro by LeRB.      *; 
*          Use at your own risk.                         *; 
*          There is an edit                              *; 
*          of your macro invocation parameters,          *; 
*          but that is no guarantee that                 *; 
*          it cannot be improperly invoked               *; 
*          with adverse consequences.                    *; 
**********************************************************; 
 
%if %length(&FtNumber) EQ 0 
  or 
    not %eval(1 LE &FtNumber LE 10) 
%then %do; 
  %put USER ERROR in of macro WebPubStdFootNote:;  
  %put FtNumber was missing or invalid; 
  %GoTo MacroExit; 
%end; 
 
footnote&FtNumber j=L 
%if %substr(&SYSVER,1,1) eq 8 
%then %do; 
h=&SAS8defaultFontSize 
%end; 
%else 
%if %substr(&SYSVER,1,1) eq 9 
%then %do; 
h=&SAS9defaultFontSize 
%end; 
'SAS and other SAS product or service names are trademarks or registered trademarks of 
SAS Institute Inc. in the USA and other countries. ' 
'® denotes USA registration.' 
j=L 
' ' 
j=L 
"bessler@execpc.com - Fox Point, Milwaukee, Wisconsin, USA - &Year." 
j=L 
'Web page created by Software User Services (SUS), not-for-profit administrator of SAS 
users conferences in the North Central USA' 
j=L 
%if %eval(1 LT &FtNumber) %then %do;  
j=L 

' ' 
%end; 
; 
 
%MacroExit: 
 
options source; 
 
%mend  WebPubStdFootNote;

The SAS code involved is rather straightforward, and I will make no attempt here to present all 
of it here or to explain it completely. If you send me an email, I will send you a zip file of the 
2009 Conference Proceedings and Tools CD. If you request the zip file and have questions, I am 
happy to answer them. 
 
In some cases, rather than being links to other web pages, the links are links to a folder of files or 
are links to individual files. 
 
All of the web packaging could be done with some tool other than SAS, but whenever SAS 
software can do a job for me, it is my first choice. 

Conclusion 
 
There is a lot more that can be done to support SAS users than to direct them to support.sas.com 
and/or the Online Doc. This paper shares with you part of the story of what I have done and puts 
real tools into the public domain for free re-use. If you would like the zip file of the Proceedings 
and Tools CD that contains SAS code which you can use to create your own local SAS user 
support intranet web site, send me an email.   

